

```

10 REM
20 REM
30 REM
40 REM
50 ON ERROR GOTO 750
60 CLS:LOCATE 1,12:PRINT "IF YOU'VE JUST LOADED,REMEMBER TO PRESS ",,,,,"STOP":FOR D=1 TO 1000:NEXT D
70 REM DEFINE CHARACTERS 203-206
80 SYMBOL AFTER 200
90 SYMBOL 203,126,255,153,255,255,129,255,126
100 SYMBOL 204,51,51,204,204,51,51,204,204
110 SYMBOL 205,128,64,240,216,252,192,128,192
120 SYMBOL 206,0,126,94,118,110,122,94,126
130 SYMBOL 207,96,145,128,95,63,63,63,96
140 MODE 1:CLS:PRINT "QUARTER ___ HALF ___ PAST ___ O'CLOCK ___ TO"
150 BORDER 0:PAPER 0:PEN 1:INK 0,11:INK 2,26
160 REM DRAW 2 CONCENTRIC CIRCLES
170 DEG
180 FOR I=420 TO 60 STEP -2
190 PLOT 176,200,1
200 PLOTR 150*COS(I),150*SIN(I):PLOT 176,200:PLOTR 170*COS(I),170*SIN(I):NEXT I
210 REM PRINT NUMBERS ON CLOCK
220 RESTORE 230:FOR I=1 TO 15:READ X,Y,C:LOCATE X,Y:PRINT CHR$(C):NEXT I
230 DATA 16,6,49,19,9,50,20,13,51,19,17,52,16,20,53,12,22,54,7,21,55,4,17,56,3,13,57,4,9,49,5,9,48,7,6,49,8,6,49,11,4,49,12,4,50
240 N=0:DIS=0:REM NO. OF TRIES & DISTANCE FROM CHEESE
250 LOCATE 26,16:PRINT SPACE$(11);CHR$(206)
260 LOCATE 23,16:PRINT SPACE$(13):PEN 2:LOCATE 26+DIS,16:PRINT CHR$(207);CHR$(205):PEN 1
270 LOCATE 12,13:PRINT CHR$(133)
280 REM WINDOWS
290 WINDOW #1,23,39,4,13:PAPER #1,14:PEN #1,3:CLS #1
300 REM DRAW RIGHT/WRONG MACHINE
310 LOCATE 28,22:PRINT CHR$(135);STRING$(7,CHR$(131));CHR$(139):LOCATE 28,23:PRINT CHR$(133);SPACE$(7);CHR$(138):LOCATE 28,24:PRINT
CHR$(141);STRING$(7,CHR$(140));CHR$(142)
320 PLOT 578,40,1:DRAW 610,72:LOCATE 39,21:PRINT CHR$(203)
330 REM PICK A RANDOM TIME
340 RANDOMIZE TIME
350 HR=INT(RND*12)+1
360 MN=INT(RND*12)*5
370 H=450-(HR*30)-(MN/2):PLOT 176,200:DRAWR 75*COS(H),75*SIN(H)
380 REM DRAW HANDS
390 M=450-((MN/5)*30):PLOT 176,200:DRAWR 100*COS(M),100*SIN(M)
400 REM COMPUTER WORKS OUT STRING FOR ANSWER
410 IF MN=0 THEN A$=STR$(HR)+" O'CLOCK":GOTO 470
420 IF MN>30 THEN B$=" TO":MN=60-MN:HR=HR+1 ELSE B$=" PAST"
430 IF HR=13 THEN HR=1
440 IF MN=15 THEN Z$=" QUARTER"ELSE IF MN=30 THEN Z$=" HALF"ELSE Z$=STR$(MN)

```

```

450 A$=Z$+B$+STR$(HR):REM A$ CONTAINS THE CORRECT ANSWER
460 REM ASKS THE TIME-TRY$ HOLDS YOUR ANSWER
470 CLS #1:PRINT #1,, "WHAT IS THE TIME?";,
480 INPUT #1,TRY$:TRY$=" "+UPPER$(TRY$):IF LEN(TRY$)<6 THEN 480
490 IF INSTR(TRY$,"O'CLOCK")+INSTR(TRY$,"TO")+INSTR(TRY$,"PAST")+INSTR(TRY$,"QUARTER")+INSTR(TRY$,"HALF") THEN 510
500 PRINT #1,, "TRY AGAIN":GOTO 480
510 REM PULL DOWN LEVER TO ACTIVATE RIGHT/WRONG MACHINE
520 PLOT 578,40,0:DRAW 610,72,0:LOCATE 39,21:PRINT " ":PLOT 578,40,1:DRAW 610,8:LOCATE 39,25:PRINT CHR$(203)
530 REM MACHINE IN ACTION
540 RESTORE 650:FOR I=1 TO 28:GOSUB 640:LOCATE 29,23:PRINT STRING$(7,206):CALL &BD19:LOCATE 29,23:PRINT STRING$(7,CHR$(204)):NEXT I
550 REM CHECK YOUR ANSWER AGAINST CORRECT ONE-RW$ HOLDS 'RIGHT' OR 'WRONG'
560 IF TRY$=A$ THEN RW$="RIGHT !":CH=224:DIS=DIS+1 ELSE RW$="WRONG !":CH=225:DIS=DIS-1
570 LOCATE 29,23:PRINT RW$:LOCATE 39,25:PRINT CHR$(CH)
580 IF TRY$<>A$ THEN PRINT #1,, "THE ANSWER WAS..." ,,,,A$
590 FOR DELAY=1 TO 5000:NEXT DELAY
600 REM ERASE HANDS & MACHINE LEVER
610 PLOT 176,200,0:DRAW 100*COS(M),100*SIN(M):PLOT 176,200:DRAW 75*COS(H),75*SIN(H):PLOT 578,40,0:DRAW 610,8:LOCATE 39,25:PRINT "
620 N=N+1:IF DIS=10 THEN 680 ELSE IF DIS=-4 THEN DIS=-3:GOTO 260 ELSE 260
630 REM HICKORY DICKORY DOCK-NO=NOTE,DU=DURATION
640 READ NO,DU:SDUND 1,NO,DU:RETURN
650 DATA 568,20,506,20,478,20,478,20,426,20,379,20,358,80,568,30,568,30,478,30,426,40,379,30,358,80,478,30,358,30,358,50,379,30,379,
50,426,50,426,30,478,40
660 DATA 478,30,426,30,478,30,536,30,568,30,638,30,716,130
670 REM MOUSE REACHES CHEESE
680 CLS:LOCATE 15,5:PRINT "WELL DONE !":LOCATE 6,10:PRINT "YOU GOT TO THE CHEESE IN";N;"GOES"
690 MSG=5+(N<25)+(N<20)+(N<15)+(N=10):RESTORE 740:FOR I=1 TO MSG:READ MSG$:NEXT I
700 LOCATE 10,15:PRINT "THAT ";MSG$
710 LOCATE 4,20:PRINT "DO YOU WANT TO START AGAIN (Y/N)"
720 K$=UPPER$(INKEY$):IF K$="Y" THEN 140 ELSE IF K$<>"N" THEN 720
730 END
740 DATA IS VERY GOOD,IS GOOD,IS AVERAGE,COULD BE A LOT BETTER,MEANS YOU NEED PRACICE !
750 PRINT "Error no. ";ERR;"detected in line";ERL
760 PRINT:PRINT "See APPENDIX VIII to see details of it.":END

```